
Обзор
IntelliSense это вероятно наилучшее, что было добавлено в Visual FoxPro. IntelliSense предоставляет высочайшую поддержку продуктивности разработчикам приложений VFP, чем что-либо добавленное ранее или с самого начала. Однако, до VFP 9, оно было ограничего окружением разработчика. Теперь IntelliSense поддерживается также и в runtime-окружении. В этом документе обсуждается, почему оно представляет полезнейшую возможность для многих типов приложений и показывается, как создать настраиваемое окружение IntelliSense непосредственно для вашего приложения.
Введение
IntelliSense это вероятно наилучшее, что было добавлено в Visual FoxPro. IntelliSense обеспечивает высочайшую поддержку продуктивности разработчикам приложений VFP, чем что-либо добавленное ранее или с самого начала. Однако, до VFP 9, оно было ограничено окружением разработчика. Начиная с VFP 9, IntelliSense поддерживается  и в runtime-окружении. До того, как мы посмотрим – как это сделать, позвольте обсудить, почему это так.

Позднее я предоставил хуки в моих приложениях, позволяющие настраивать их. Например, в Stonefield Query пользователь может указать код, который может быть исполнен в ряде мест: при старте приложения и при его закрытии, после загрузки словаря данных (с тем, чтобы он, пользователь, смог динамически менять их при необходимости), до входа пользователя в систему, после получения данных по запросу, но до выпуска отчета, после запуска отчета и тому подобного. Причиной этому является гибкость; я не  имею возможности приходить при каждом изменении конфигурации к каждому пользователю, поэтому я позволяю им делать это самим. Очевидно, это требует знания языка VFP, но это не столь затруднительно для разработчиков, персонала отделов IT или знающих пользователей, особенно в случае простых изменений.

В версии VFP 8 пользователь мог напечатать необходимый код в элементах управления типа “editbox” предоставляемых в соответствующих диалогах редактирования кода, но при этом не мог использовать преимущества IntelliSense. В VFP 9 имеется не только IntelliSense для команд и функций VFP, здесь, кроме всего прочего IntelliSense работает с объектной моделью Stonefield Query. Рисунок 1 показывает пример -  как смотрится ввод кода в VFP 9. Заметьте, что здесь, в приведенном примере, имеется как раскрашивание кода, так и список членов, отображаемый IntelliSense.

Рис 1. IntelliSense в runtime полезно в случае, если вы позволяете пользователю писать свои скрипты в ваших приложениях.

В своей статье в сентябре 2004 в журнале FoxPro Advisor под заголовоком “Take Advantage of IntelliSense in VFP 9 Applications,” Toni Feltman представил несколько других, менее технических, использований IntelliSense в runtime, таких, как предоставление нечто подобного возможности Microsoft Office’s AutoCorrect, а также IntelliSense для деловой информации, такой, медицинские диагнозы.

В этом документе, я рассматриваю два аспекта использования IntelliSense внутри приложений: внедрение IntelliSense в runtime и создание настраиваемой IntelliSense для ваших приложений.
Установка IntelliSense для Runtime
IntelliSense работает с помощью запуска программы, определенной в системной переменной _CODESENSE и использует таблицу, определенную в системной переменной _FOXCODE. По умолчанию, в среде разработки, _CODESENSE указывает на приложение FOXCODE.APP в программном каталоге VFP, а _FOXCODE указывает на таблицу FOXCODE.DBF в каталоге, определяемом HOME(7). Оба из них пусты по умолчанию в runtime окружении. Одной из задач, необходимых для предоставления возможностей IntelliSense в runtime, является установка этих системных переменных в подходящие значения в вашем коде и предоставления пользователю приложения IntelliSense и таблицы, которую это приложение использует.

Хотя я мог поставлять FOXCODE.APP совместно с  моим приложением, я в действительности желал настраиваемого IntelliSense приложения, которое поддерживало бы возможности настройки, которые я рассмотрю позднее в этом документе. Итак я смотрел, что требуется для того, чтобы создать его. Взглянув на исходный код FOXCODE.APP в программном каталоге VFP  Tools\XSource\VFPSource\FoxCode (создаваемом при распаковке XSOURCE.ZIP в Tools\XSource), я заметил, что FOXCODE.PRG является главной программой для FOXCODE.APP. Моей первой мыслью было просто включить FOXCODE.PRG в мое приложение и установить системную переменную _CODESENSE в “FOXCODE.PRG.” К несчастью, IntelliSense не работало, когда я сделал это, вероятно потому, что VFP не мог найти внутри запущенного EXE указанную программу. Поэтому я решил создать вместо этого свой собственный FOXCODE.EXE.

Для создания этого EXE, я для скопировал FOXCODE.PRG и FOXCODE.H из Tools\XSource\VFPSource\FoxCode. (Поскольку мне нужно было внести некоторые изменения в PRG, я скопировал его и сделал изменения в копии вместо того, чтобы просто ввести его, как программу, в мой проект.) Далее, я закомментировал вызовы DO FORM в главном коде и методе GetInterface в программе  FOXCODE.PRG потому, что в противном случае эти вызовы могли бы заставить project manager вытащить в проект какое-то число файлов, которые нам не требуются в runtime. Затем я создал проект с именем FOXCODE.PJX и добавил настроенный мною FOXCODE.PRG, файл CONFIG.FPW, который просто имел настройку RESOURCE=OFF (это предотвращает создение FOXUSER.DBF в случае, когда пользователь случайно запустит FOXCODE.EXE) и FOXCODE2.DBF из Tools\XSource\VFPSource\FoxCode (поскольку она используется в коде программы FOXCODE.PRG). И в завершении я собрал FOXCODE.EXE из этого проекта.

Еще одна вещь, которую я пробовал, и которая не работала: включение таблицы IntelliSense table, FOXCODE.DBF, в FOXCODE.PJX и/или проект для моего приложения. В обоих случаях, IntelliSense давало ошибку, что таблица IntelliSense не найдена. Поэтому, вы должны поставлять совместно со своим приложением FOXCODE.DBF и FPT в виде отдельных файлов. Вы можете переименовать их в какое-то другое имя; просто измените системную переменную _FOXCODE в значение, которое укажет на новое имя вашего файла таблицы. Кроме того, вам не нужно предоставлять копию таблицы IntelliSense, которую вы использовали в окружении разработки; вы можете сделать ее копию и редактировать ее содержимое по своим надобностям, например удалив записи IntelliSense для определенных команд и функций, которые не делают погоды в  runtime-окружении или введя какие-то относящиеся к делу записи для коротких ключей в мемо-полях.

Теперь у меня были настроенное приложение IntelliSense и таблица, все что требовалось сделать, так это сказать моему приложению об установках _CODESENSE и _FOXCODE в подходящие значения. Здесь представлен код из STARTUP.PRG, главной программы для проекта SAMPLE.PJX. Отметьте, что она сохраняет и восстанавливает значения _CODESENSE, _FOXCODE и путь, если мы запускаем приложение в окружении разработчика, с тем, чтобы не смешивать эти установки во время тестирования.

* Установка IntelliSense.

local lcCodeSense, ;

  lcFoxCode, ;
  lcCaption, ;

  lcPath

if version(2) = 2

  lcCodeSense = _codesense

  lcFoxCode   = _foxcode

  lcCaption   = _screen.Caption

  lcPath      = set('PATH')

  set path to SOURCE

  push menu _msysmenu

endif version(2) = 2

_codesense = 'FoxCode.EXE'

_foxcode   = 'FoxCode.DBF'

* здесь выполняются нормальные задачи приложения.

_screen.Caption = 'Sample Application'
do SampleMenu.mpr
read events

* Если мы находимся в режиме разработки, восстанавливает настройки IntelliSense, путь,
* заголовок и меню до выхода из программы.

if version(2) = 2

  pop menu _msysmenu

  _codesense      = lcCodeSense

  _foxcode        = lcFoxCode

  _screen.Caption = lcCaption

  set path to &lcPath

endif version(2) = 2

Использование IntelliSense в Runtime

Равно, как и в design time, имеется два места, в котором IntelliSense поддерживается в runtime: в окне редактирования файла кода программы (PRG) и в мемо-поле. Но было бы замечательно, если бы IntelliSense поддерживался в editbox из-за возможностей управления таким элементом, но у нас нет такого счастья. Отметьте, что для того, чтобы заставить IntelliSense работать в мемо-поле, нужны некоторые предпосылки: вам необходимо убедиться, что выключен перенос слов (word wrap) и включена раскраска символов кода, что обеспечивается пользовательским файлом ресурса FOXUSER с помещенным в него установками. Вследствие этого, моим предпочтением является использование PRG. Даже если то, что пользователь в конечном счете печатает находится в мемо-поле,  у вас остается возможность скопировать содержимое мемо-поля в файл, использовать окно редактирования PRG для модификации такого файла и, затем, скопировать содержимое файла назад в мемо-поле. Это как раз то, что делает приложение примера.

Метод Click кнопки Edit Code в ScriptEditor.SCX, одной из форм приложения примера, имеет приведенный ниже код:

local lcPath, ;

  loForm

* Записываем текущее содержание CODE в файл.

lcPath = addbs(sys(2023)) + trim(NAME)
strtofile(CODE, lcPath)

* Создем форму с определенными характиристиками для окна
* редактирования PRG.

loForm = createobject('Form')

with loForm

  .Caption  = trim(NAME)

  .Width    = _screen.Width  - 50

  .Height   = _screen.Height - 50

  .FontName = 'Courier New'

  .FontSize = 10

endwith

* Редактируем code в окне PRG, затем возвращаем результат
* обратно в CODE.

modify command (lcPath) window (loForm.Name)

replace CODE with filetostr(lcPath)

erase (lcPath)

Этот код, записывает текущее содержание записи в PRG в каталоге временных файлов пользователя, создает форму с характеристиками, которые нам нужны для окна редактирования, редактируем файл и, затем помещает его содержимое обратно в таблицу и удаляем PRG. Рисунок 2 показывает, как выглядит IntelliSense, когда вы запускаете приложение примера, выбираете скрипт и щелкаете на командной кнопке.

Рисунок 2. IntelliSense работает в runtime так же хорошо, как и в design time.

Отметьте, что над определенными вещами, такими как шрифт, размер окна и его размещение, у вас нет достаточного контроля, когда мы говорим об окне, в котором появляется PRG. Одним из нежелательных поведений является то, что щелчок вне окна редактирования кода вызывает его автоматическое закрытие (пока вы не используете NOWAIT, которое влечет за собой другие сложности). Кроме того, если ваше приложение представляет собой top-level форму, то здесь имеются и другие проблемы. Если вы не используете опцию IN WINDOW для предложения MODIFY COMMAND и указываете имя вашей top-level формы, а _SCREEN невидим, что обычно не так, когда вы используете top-level форму, окно редактирования кода вообще не появляется. Использование IN WINDOW делает окно редактирования кода дочерним окном top-level формы, вследствие чего оно не может быть перемещено из формы или не может быть изменено в размерах таким образом, что станет больше родительской формы. Поэтому, я опасаюсь, что нам просто остается жить с этим окном с потерей управления, которое мы обычно имеем над ним.

Конечно, IntelliSense может быть использован шире, чем просто для редактирования кода; это еще и полезная вешица для расширения текста, подобно возможности AutoCorrect feature в Microsoft Word. Например, в приложении по учету рабочего времени, описания вводимые в рабочие карточки часто содержат часто используемые предложения, наподобие  “Работает над ” или “Встречается с.” Было бы неплохо, чтобы пользователь мог напечатать сокращение и оно автоматически превратилось бы в полный желаемый текст. Это легко сделать с IntelliSense: создайте запись с TYPE = “U,” ABBREV = используемому сокращению, а EXPANDED содержит полный текст для данного сокращения. Более того и даже лучше, вы можете иметь список значений для выбора из всплывающего меню с помощью указания “{cmdhandler}” в поле CMD и списка значений для отображения, каждое из которых должно размещаться на собственной строчке в поле DATA. Рисунок 3 показывает, что происходит, когда пользователь печатает “wo” и нажимает клавишу пробела в окне редактора для Comment на форме SampleForm.SCX.

Рисунок 3. IntelliSense позволяет не только заменять сокращения полным текстом, он позволяет включить список дополнительных значений.

Проблемы с IntelliSense

Хотя IntelliSense является величайшей вешью после нарезанного хлеба, была одна вешь, которая раздражала меня, а именно то, что при использовании IntelliSense в классе, отображались все члены классы, хотя я предпочтел бы видеть только те, которые я действительно желаю видеть.

Например, Рисунок 4 показывает отображение IntelliSense для класса ConnectionMgr. Отметьте, что здесь только есть несколько пользовательских свойств и методов, которые действительно нужны сейчас, IntelliSense отображает все. Это требует дополнительных усилий для выбора интересующего вас члена, особенно если вы не очень знакомы с самим классом.

Рисунок 4. Хотя IntelliSense позволяет вам выбрать имя члена из списка, но в то же время отображат больше элементов, чем вам реально необходимо.

У нас была схожая проблема с окном Properties до выхода VFP 9. В этой версии Microsoft ввел закладку Favorites, которая отображает только тех членов, которые вы определили, чтобы они появились на этой закладке. (Один из способов сделать это заключается в вызове контекстного меню (щелчок по правой клавише мыши) на имени члена и выбора из него пункта Add to Favorites.) Итак, вместо того, чтобы пробираться через дюжины членов с целью найти требуемого, вы можете просто выбрать его среди тех немногих, которых вы определили для появления на закладке Favorites.

То, что я был бы действительно рад увидеть, так это эквивалент закладки Favorites для IntelliSense. Поэтому я решил создать это. Результатом явились Favorites for IntelliSense (FFI).

Фавориты для IntelliSense
До того, как мы заглянем под покрывало, посмотрите FFI в действии. Первой вешью, которую нам необходимо сделать,  является регистрация класса с FFI. Откройте класс ConnectionMgr в библиотеке ConnMgr.VCX (включена в исходники, сопровождающий этот документ), затем напечатайте в командном окне DO FORM FFIBUILDER. Вы должны увидеть диалог, показанный на рисунке 5. Этот диалог позволит вам указать “namespace”-пространство имен – для класса. Namespace представляет собой имя, которое покажется в списке IntelliSense, когда вы напечатаете “LOCAL какя-то переменная AS”. По умолчанию оно равно имени класса, но вы можете указать что-либо другое на ваше усмотрение. Например, для класса с именем cApplication, вы можете пожелать, чтобы пространство имен было определено как Application.


Описание используется в качестве tooltip для класса в списке IntelliSense, если этот класс используется, как член другого класса (например, свойство User класса примера cApplication содержит экземпляр класса cUser, поэтому описание класса используется для свойства User). По умолчанию оно равно описанию класса, так как оно сделано в описании Class Info в меню Class или с помощью  Edit Description в меню Project, доступном, когда класс выбран в Project Manager.

Рисунок 5. Построитель Favorites for IntelliSense позволяет вам указать, что именно появится в IntelliSense для выбранного класса.

Элемент TreeView в области Members показывает глобальные пользовательские свойства и методы класса; если вы желаете, чтобы наряду с ними отображались “родные” члены класса, измените предложение AMEMBERS() в методе LoadTree класса FFIBuilderForm в библиотеке FFIBuilderForm.VCX. Чек-бокс, размещенный слева от имени члена указывает – будет ли этот член класса включен в IntelliSense или нет; по умолчанию в отображение включены все пользовательские члены.

Описание используется в качестве tooltip для члена в списке IntelliSense и по умолчанию равняется описанию, введенному вами для него при его создании. То, что помещено в “method signature” отображается как tooltip для метода, когда вы при печати в окне редактирования открываете скобку или печатаете запятую в списке параметров метода. Этот tooltip покажет вам, какие параметры могут быть посланы методу. “signature” (далее в переводе – сигнатура метода) по умолчанию равна имени метода, а содержание - любому предложению LPARAMETERS в коде метода, но вы можете отредактировать его для отображения чего-либо еще желаемого, включая тип данных  возвращаемого значения.

В боксе редактирования “script” вы можете ввести код, который будет исполнен при выборе этого члена и печати открывающей скобки, запятой в списке параметров или знака равенства (для присвоения значения свойству). Мы посмотрим этот пример позже.

Сделайте любые изменения в этом диалоге и затем щелкните на OK. Эта форма добавит записи для класса и каждого из зарегистрированных членов в таблицу FFI. Это, кроме того, добавит при необходимости две записи в вашу таблицу IntelliSense: одну для класса и одну для скрипта IntelliSense для FFI. Мы также посмотрим на это более пристально попозже.

Для того, чтобы увидеть, как работает Favorites for IntelliSense с этим классом, создайте PRG и напечaтайте LOCAL ConnMgr AS. Когда вы нажмете на клавишу пробела после печати AS, вы должны увидеть список типов IntelliSense, такой как показан на рисунке 6. Когда вы выберите ConnectionMgr и нажмете Enter, вы увидите приведенный ниже код, вставленным в PRG (где Path представляет собой путь к библиотеке VCX):

local ConnMgr as ConnectionMgr

ConnMgr = newobject('Connectionmgr', 'Path\connmgr.vcx')

Рисунок 6. Любой класс, зарегистрированный с Favorites for IntelliSense отображается как тип в IntelliSense.

Теперь напечатайте ConnMgr и завершите печать точкой. Как вы можете видеть это в рисунке 7, IntelliSense отобразит только зарегистрированных членов класса (отсюда и идет - “Favorites for IntelliSense”). Если вы выберите метод, такой как ExecuteSQLStatement, то, когда вы напечатаете открывающую скобку, вы увидите опреленную вами сигнатуру метода в виде IntelliSense tooltip, что значительно упрощает видение – какие параметры должны быть посланы методу.

Рисунок 7. IntelliSense отображает только зарегистрированных членов указанного класса. Сравните это с рисунком 4.

Как работает FFI

Секрет, заключенный в FFI определяется двумя вещами: как IntelliSense обращается с вещами, определенными, как “types” в таблице IntelliSense и в скриптах IntelliSense. Типы, обычно, используются для типов данных (таких как Integer или Character) и базовых классов (таких как Checkbox и Form). Однако, вы можете определить в качестве типов и другие вещи, либо вручную вводя записи с полем TYPE установленным в “T” или с помощью IntelliSense Manager, доступного в меню Tools.


Другой вид записей типа обычно представляют пользовательские классы или объекты COM, что позволяет заполучить для них IntelliSense. That’s what we’ll use type records for as well, but we’ll customize how IntelliSense works by using a script and a custom IntelliSense-handling class.

Если вы посмотрите на вашу таблицу IntelliSense (с помощью USE (_FOXCODE) AGAIN и BROWSE) после использования формы FFIBUILDER для регистрации класса, вы увидите две новые записи в конце таблицы. Одна представляет запись типа для класса; она не содержит значительной информации, а только пространство имен, определенное вами в форме построителя и имя скрипта, который будет использован IntelliSense для класса; это определено как “{HandleFFI}” в колонке CMD. Другая запись представляет собой запись скрипта, для которой поле TYPE установлено в “S”, а поле ABBREV содержит “HandleFFI,” имя, указанное в колонке CMD первой записи.

Запись скрипта в ее поле DATA имеет приведенный ниже код (в этом коде, Path1 замещается путем к библиотеке FFI.VCX, а Path2 замещается с путем к приложению FoxCode.EXE):

lparameters toFoxCode

local loFoxCodeLoader, ;

  luReturn

if file(_codesense)

  set procedure to (_codesense) additive

  loFoxCodeLoader = createobject('FoxCodeLoader')

  luReturn        = loFoxCodeLoader.Start(toFoxCode)

  loFoxCodeLoader = .NULL.

  if atc(_codesense, set('PROCEDURE')) > 0

    release procedure (_codesense)

  endif atc(_codesense, set('PROCEDURE')) > 0

else

  luReturn = ''

endif file(_codesense)

return luReturn

define class FoxCodeLoader as FoxCodeScript

  cProxyClass    = 'FFIFoxCode'

  cProxyClasslib = 'Path1\FFI.vcx'

  cProxyEXE      = 'Path2\FoxCode.EXE'

  procedure Main

    local loFoxCode, ;

      luReturn

    if version(2) = 2

      loFoxCode = newobject(This.cProxyClass, This.cProxyClasslib)

    else

      loFoxCode = newobject(This.cProxyClass, ;

        juststem(This.cProxyClasslib), This.cProxyEXE)

    endif version(2) = 2

    if vartype(loFoxCode) = 'O'

      luReturn = loFoxCode.Main(This.oFoxCode)

    else

      luReturn = ''

    endif vartype(loFoxCode) = 'O'

    return luReturn

  endproc

enddefine
Этот код определяет субкласс класса FoxCodeScript, определенного в приложении IntelliSense, указанном в системной переменной _CODESENSE. Этот субкласс переписывает метод Main (который вызывается IntelliSense) для создания экземпляра класса FFIFoxCode из FFI.VCX и вызывает его метод Main. Вызов посылает ссылку на объект данных IntelliSense, который содержит информацию о том, что напечатал пользователь и прочие установки IntelliSense. Отметьте, что FFIFoxCode создается либо из FoxCode.EXE, либо из самостоятельной библиотеки VCX, в зависимости от того, где мы находимся – в окружении runtime или окружении разработки; мы не хотим использовать FoxCode.EXE в разработке окружения потому, что оно имеет таблицу  с именем FFI.DBF, о которой я коротко расскажу, встроенную в приложение, а мы не хотим копировать таблицу в этом случае,  поскольку она может содержать устаревшие данные.

Скрипт HandleFFI используется для всех классов, зарегистрированных с FFI, поэтому для этого скрипта в таблице имеется только одиночная запись.

В результате исполнения этого скрипта, для всех задач IntelliSense для любого зарегистрированного в FFI пространства имен вызывается FFIFoxCode. Main вызвается для всех задач IntelliSense для любого зарегистрированное в FFI пространства имен, например, когда вы выбираете пространство имен из списка отображаемого IntelliSense, когда вы печатаете “LOCAL какая-то переменная AS” или когда вы печатаете один из вызывающих “срабатывание” символов—таких, как точка, открывающая скобка или знак равнства—в предложении, содержащем имя переменной, в которую создается экземпляр класса.

FFIFoxCode

Класс FFIFoxCode выполняет всю настроенную работу IntelliSense для FFI, поэтому давайте разберем подробно этот класс.

Метод Init method выполняет всего две вещи: 

· Включает отладку в системных компонентах (без этого, вы не сможете легко отловить проблемы в коде)

Открывает FFI.DBF (таблицу, содержащую регистрационную информацию о классе и его членах) вызовом метода OpenFFITable. Если таблица не может быть открыта, Init отображает сообщение об ошибке и возвращает .F. для того, чтобы экземпляр класса не был создан.

* Включаем отладку.

sys(2030, 1)

* Открываем таблицу FFI (Favorites for IntelliSense).

local llReturn

llReturn = This.OpenFFITable()

if not llReturn

  messagebox('Не могу открыть таблицу Favorites for IntelliSense.', 64, 'Обработчик IntelliSense')

endif not llReturn

return llReturn

Метод Main, вызываемый из скрипта IntelliSense, обрабатывает все задачи IntelliSense для FFI. Как мы видели ранее, скрипт посылает объект FoxCode методу Main. Свойство Data этого объекта содержит содержимое мемо-поля Data соответствующей записи в таблице FoxCode, которой случилось быть записью для зарегистрированного класса, в том случае, когда Data содержит пространство имен класса. Main вызывает метод GetFFIMember для определения, какой член класса был напечатан вами (это может быть, кроме того, и класс сам по себе) и возвращает объект SCATTER NAME из соответствующей записи таблицы FFI.

Если пространство имен найдено в свойстве MenuItem объекта FoxCode (MenuItem содержит подсказку элемента, выбранного пользователем, если IntelliSense был вызван выбором элемента из меню IntelliSense), мы должны быть на предложении “LOCAL какая-то переменная AS”, так как Main вызывает метод HandleLOCAL для работы с предложением. В противном случае, Main выполняет одну из нескольких вещей. Если свойство MenuItem объекта FoxCode содержит значение, то есть пользователь выбрал значение из списка для присвоения значения свойству, то он вызывает метод HandlePropertyAssignment для вставки выбранного значения в код (мы увидим как это используется чуть позже). Если символ, который вызвал срабатывание IntelliSense представляет собой точку, нам необходимо отобразить список зарегистрированных членов класса, поэтому Main вызывает DisplayMembers для того, чтобы выполнить эту работу. Есди символ, который вызвал срабатывание IntelliSense представляет собой открывающую скобку, знак равенства или запятую и для этого случая имеется код в мемо-поле SCRIPT записи FFI, то этот код исполняется. Этот код скрипта мог бы, например, указать список перечисляемых значений, который IntelliSense мог бы отобразить для значения свойства, или параметр метода (мы увидим пример этого позже).

Наконец, если мемо-поле TIP записи FFI заполнено, Main использует его содержимое в качестве tooltip для IntelliSense. Это обычно используется для сигнатуры метода (например, “Login(UserName as String, Password as String) as Boolean”).

* Это главная процедура (main), которая вызывается из скрипта IntelliSense для класса FFI
lparameters toFoxCode

local lcNameSpace, ;

  loData, ;

  lcReturn, ;

  lcTrigger

with toFoxCode

* Получаем пространство имен и объект из таблицы FFI для этого пространства имен. 
* Кроме того, получаем символ, который “запускает” IntelliSense.

  lcNameSpace = .Data

  loData      = This.GetFFIMember(.UserTyped, lcNameSpace)

  lcReturn    = ''

  lcTrigger   = right(.FullLine, 1)

  do case

* Пользователь выбрал элемент из списка для присвоения свойству.

    case vartype(loData) <> 'O' and not empty(.MenuItem)

      lcReturn = This.HandlePropertyAssignment(toFoxCode)

* Мы не можем найти члена в таблице FFI, поэтому ничего не делаем.

    case vartype(loData) <> 'O'

* Если мы находимся в предложении LOCAL, обработаем это возвратом текста
* который мы желаем вставить.

    case atc(lcNameSpace, .MenuItem) > 0

      lcReturn = This.HandleLOCAL(toFoxCode, lcNameSpace, ;

        trim(loData.Class), trim(loData.Library))

* Если мы были “запущены” посредством ".", отображаем список членов.

    case lcTrigger = '.'

      This.DisplayMembers(toFoxCode, loData)

* Если мы были “запущены” посредством "(" (для запуска метода вывода списка 

* параметров),  "=" * (для свойства)  или "," (для ввода нового параметра) 
* и у нас имеется для этого скрипт, то исполняем его.

    case inlist(lcTrigger, '=', '(', ',') and not empty(loData.Script)

      lcReturn = execscript(loData.Script, toFoxCode, loData)

* Если мы были “запущены” посредством "(" или "," , отображаем tooltip для метода
* (обычно сигнатуру метода).

    case inlist(lcTrigger, '(', ',') and not empty(loData.Tip)

      .ValueTip  = loData.Tip

      .ValueType = 'T'

  endcase

endwith

return lcReturn

Мы не будем рассматривать здесь код для HandleLOCAL; вы можете сами исследовать его. Main вызывает его, когда вы печатаете “LOCAL какая-то переменная AS” и выбираете одно из зарегистрированных пространств имен из списка типов. Все что он делает заключается в генерации предложения NEWOBJECT() для класса,  пока вы его не имеете. Единственной сложностью для этого метода является определение регистра (здесь имеются в виду большие или маленькие буквы) используемого для NEWOBJECT().

(Я мог бы жестко прописать это в виде “newobject” так как я всегда использую нижний регистр для ключевых слов FoxPro, но решил быть дружественным для других разработчиков, у которых свои препочтения  в регистрах. Это решено через поиск в таблице IntelliSense записи с TYPE = “F” (для “function”) и ABBREV = “NEWO” (сокращением для NEWOBJECT) и использованием значения в поле CASE.)

GetFFIMember, вызывамый из Main, ищет класс или члена, которого вы напечатали, в таблице FFI. Он использует свойство UserTyped объекта FoxCode (посылаемого в качестве первого параметра), которое содержит напечатанный вами текст, имеющий отношение к пространству имен. Например, когда вы печатаете “llStatus = ConnMgr.ExecuteSQLStatement(”, UserTyped содержит “ConnMgr.ExecuteSQLStatement”.

 GetFFIMember начинает с поиска записи для указанного класса в таблице FFI. Если в UserTyped включена точка, то затем он ищет соответствующую запись члена. Если он находит корректную запись, он возвращает объект SCATTER NAME из этой записи.

* Определение, какого члена пространства имен напечатал пользователь и возврат объекта SCATTER
* NAME из соответствующей записи таблицы FFI.

lparameters tcUserTyped, ;

  tcNameSpace

local loReturn, ;

  lcUserTyped, ;

  llFound, ;

  lnPos, ;

  lcMember, ;

  lnSelect

* Забираем то, что напечатал пользователь. Если печать заканчивается открывающей скобкой, 

* отрезаем ее.

loReturn    = .NULL.

lcUserTyped = alltrim(tcUserTyped)

if right(lcUserTyped, 1) = '('

  lcUserTyped = substr(lcUserTyped, len(lcUserTyped) - 1)

endif right(lcUserTyped, 1) = '('

* Ишем запись для класса в таблице FFI. Если в напечатанном тексте имеется точка,  
* пытаемся найти запись для члена.

if seek(upper(padr(tcNameSpace, len(__FFI.CLASS))), '__FFI', 'MEMBER')

  llFound = .T.

  lnPos   = at('.', lcUserTyped)

  if lnPos > 0

    lcMember = alltrim(__FFI.MEMBER) + substr(lcUserTyped, lnPos)

    llFound  = seek(upper(padr(lcMember, len(__FFI.MEMBER))), '__FFI', ;

      'MEMBER')


endif lnPos > 0

* Если мы нашли требуемую запись, создаем для нее объект SCATTER NAME.

  if llFound

    lnSelect = select()

    select __FFI

    scatter memo name loReturn

    select (lnSelect)

  endif llFound

endif seek(upper(padr(tcNameSpace ...

return loReturn

DisplayMembers, вызывается из Main для того, что сказать  IntelliSense что оно должно отобразить список зарегистрированных членов для класса, когда вы печатаете точку в командной строке. DisplayMembers вызывает GetMembers для извлечения коллекции членов для указанного класса (мы не будем рассатривать здесь этот метод). Затем он заполняет массив Items объекта FoxCode именами и описаниями членов и устанавливает свойство ValueType объекта в “L,”, что сообщает IntelliSense о необходимости отобразить listbox с содержимым массива Items.

Этот код показывает игнорирование одной ошибки разработки IntelliSense: объект FoxCode имеет одиночное свойство Icon, которое содержит имя файла картинки, отображаемой в listbox. Что является реально необходимым, так это наличие дополнительной колонки в массиве Items, так как в этом случае мы смогли отображать разные картинки для свойств и методов. К несчастью, мы можем взять только одну картинку, отображаемую для всех элементов списка.

* Строим список членов для отображения IntelliSense.

lparameters toFoxCode, ;

  toData

local loMembers, ;

  lcPath, ;

  lnI, ;

  loMember

with toFoxCode

* Получаем коллекцию членов для текущего класса.

  loMembers = This.GetMembers(alltrim(toData.Member))

  if loMembers.Count > 0

* Добавляем каждый член в массив Items объекта FoxCode.

    dimension .Items[loMembers.Count, 2]

    lcPath = iif(file('propty.bmp'), '', home() + 'FFC\Graphics\')

    for lnI = 1 to loMembers.Count

      loMember       = loMembers.Item(lnI)

      .Items[lnI, 1] = loMember.Name

      .Items[lnI, 2] = loMember.Description

      if loMember.Type = 'P'

        .Icon = lcPath + 'propty.bmp'

      else

        .Icon = lcPath + 'method.bmp'

      endif loMember.Type = 'P'

    next loMember

* Устанавливаем свойство ValueType объекта FoxCode в значение "L", означающее
* необходимость отображения элемента listbox, содержащего элементы, 
* определенные в массиве Items.

    .ValueType = 'L'

  endif loMembers.Count > 0

endwith

Некоторые успешные использования
Имеется несколько интересных вещей, которые вы можете делать с FFI для более легкого его использования. Одна из них заключается в том, что классы могут содержать иерархию объектов, и с помощью регистрации классов, содержащихся в качестве членов класса, в FFI, вы можете получить ясный IntelliSense для этих членов. Например, класс cApplication создает экземпляр класса cUser в своем свойстве User в runtime. Обычно, вы получили бы IntelliSense на свойстве User, потому, что оно содержит объект в design time.Однако, если вы откроете класс cUser и укажите “Application.User” в качестве пространства имен, вы получите IntelliSense для члена User пространства имен Application, так как это показано на рисунке 8.

Рисунок 8. Вы даже можете получить IntelliSense на объектах, созданных в runtime, если они зарегистрированы в субпространстве имен главного пространства имен.

Другой полезной вешью является написание скрипта, определяющего то, как IntelliSense обращается с членами при вводе кода в editbox области Script формы построителя FFI. Например, свойство DatabaseSpecifier в ConnectionMgr указывает, как могла бы быть указана база данных указана в строке соединения ODBC. Для некоторых баз данных, таких как Access и dBase, вы указываете базу данных, используя “dbq=database path”, тогда как для других вы указываете “database=database name.”

Вместо того, чтобы предполагать, что разработчик знает, что действительным выбором для DatabaseSpecifier являются “database” и “dbq” (и печати  их без каких-либо ошибок), вы можете приказать IntelliSense отобразить список действительных значений (как показано на рисунке 9) в схожей манере пути отображения указанных членов списка: с помощью заполнения массива Items объекта FoxCode и установить его свойство ValueType в “L.” Код скрипта для DatabaseSpecifier делает именно это.

lparameters toFoxCode, ;

  toData
dimension toFoxCode.Items[2, 2]

toFoxCode.Items[1, 1] = 'database'

toFoxCode.Items[1, 2] = 'Used for most databases'

toFoxCode.Items[2, 1] = 'dbq'

toFoxCode.Items[2, 2] = 'Used for Access and dBase'

toFoxCode.ItemScript  = 'HandleFFI'

toFoxCode.ValueType   = 'L'

return ''

Рисунок 9. Используя скрипт, вы можете отобразить список пользовательских значений для присвоения свойству..

Заметьте, что все скрипты посылают ссылку на объект FoxCode и объект SCATTER NAME для записи FFI. Кроме того, отметьте, что этот код устанавливает ItemScript в “HandleFFI;” который говорит IntelliSense, что нужно вызвать скрипт HandleFFI script, который в свою очередь вызывает FFIFoxCode.Main, после того, как пользователь выберет значение. Как мы видели ранее, Main вызывает HandlePropertyAssignment, когда свойство MenuItem объекта FoxCode содержит значение, что происходит в случае, когда пользователь выбирает значение из предоставляемого нами списка. Этот метод заключает значение  в кавычки, если оно не является численным; без которых значение не будет корректно работать. Наконец, отметьте, что возвращамое скриптом значение посылается IntelliSense, так что в случае, когда вы не желаете вставлять значение в код, возвратите пустую строку и установите поле ValueType в что-нибудь отличное от “V.”

Здесь приведен скрипт, который отображает контекстно-чувствительный tooltip для некоторых параметров метода и списка значений одного из них:

lparameters toFoxCode, ;

  toData

local lcMember, ;

  lnPos, ;

  lcParameters, ;

  lnParameters

lcMember     = alltrim(substr(toData.Member, rat('.', toData.Member) + 1))

lnPos        = atc(lcMember, toFoxCode.FullLine)

lcParameters = substr(toFoxCode.FullLine, lnPos + len(lcMember))

lnParameters = occurs(',', lcParameters) + 1

do case

  case lnParameters = 1

    toFoxCode.ValueTip  = 'First parameter tooltip'

    toFoxCode.ValueType = 'T'

  case lnParameters = 2

    toFoxCode.ValueTip  = 'Second parameter tooltip'

    toFoxCode.ValueType = 'T'

  case lnParameters = 3

    dimension toFoxCode.Items[2, 2]

    toFoxCode.Items[1, 1] = '1'

    toFoxCode.Items[1, 2] = 'Description for value 1'

    toFoxCode.Items[2, 1] = '2'

    toFoxCode.Items[2, 2] = 'Description for value 2'

    toFoxCode.ValueType = 'L'

endcase

Обновление приложения IntelliSense

Для поддержки FFI, введите FFI.VCX и FFI.DBF в FOXCODE.PJX. Кроме того, я включил файлы картинок,

которые FFI использует для списка членов, PROPTY.BMP и METHOD.BMP, в FOXCODE.PJX но список членов не имеет картинок. Я нашел, что эти файлы картинок либо должны быть включены в главный проект приложения  (в этом случае SAMPLE.PJX) или предоставлены в виде отдельных файлов.

Проблемы
Я встретился во время тестирования с тем, что смотрится, как несколько багов или как ошибки дизайна в IntelliSense:

· Список типов, который нормально появляется, когда вы печатаете “LOCAL какая-то переменная AS” не показывается в runtime.

Завершение скобок не работает в runtime.
Если у вас имеется объект, который является членом другого объекта (такого, как член User объекта cApplication в примере этого месяца, который содержит ссылку на объект cUser), IntelliSense, который появляется для этого члена, на самом деле предназначен для родительского объекта. Например, если вы печатаете ÄLOCAL loUser as Application.User” (класс cApplication зарегистрирован как пространство имен Application), то когда вы печатаете “loUser” и точку, то IntelliSense отображает список членов для класса cApplication, а не для класса cUser.
Итоги
IntelliSense являтся одной из тех вещей, без которых мы, разработчики VFP не можем жить, однажды попробовав использовать их. Теперь вы можете дать своим пользователям теже преимущества в вашмс приложениях. Обычно этот метод не для всех, но если если пристроите его в своем приложении, я гарантирую, что ваши пользователю полюбят вас за это. Благодарю Randy Brown, бывшего Microsoft Program Manager for Visual FoxPro, указавшего мне путь для конфигурирования IntelliSense для отображения только необходимых вам членов.

